A paradox example, right simulation solutions on EcosimPro

Sommaire

1. The ideal model

A simple set of two capacitors is used: one capacitor C1 shall charge the other capacitor C2.

First to charge C1, one use a constant DC supply and a resistor R0: the switch SW1 is closed for a while. Then SW1 is open again at Time=2s. So C1 is isolated and fully charged for Time=2s to 3s.

The well known paradox is that when C1 is discharged into C2 (by closing SW1_2 at Time=3s) **half the initial energy into C1 is lost,** the total remaining energy in the circuit is half the C1 initial energy.





It is interesting to see that during the charge of C1, the energy (lost by Joule effect) into the resistor (whatever its resistance) and the one stored into the capacitor C1 (=1/2 CV²) are the same and equal to 0.1 J.

The voltage reached by C1 is 10 V.

So when C1 is discharged into C2 the voltage becomes 5 V for both as expected for a constant charge (=CV).

But more interesting is that when the switch SW1_2 closes the energies into C1 and C2 become equal and each one is a quarter of the initial of C1 : each is 0.025 J. **That is the paradox** there is once again a **loss of half the energy transferred from C1 to C2**. In the model there are no **apparent** damping, no parts for explaining that (it has been proven in the literature that the lost energy is converted into radiated EM power for ideal wires).

However, as pointed out by EcosimPro team, the ideal switch is not fully ideal because it remains a very small resistance (Roff) when closed. When computing the joule effect losses in the switch, it is found to be exactly equal to the missing energy (with 0.05 J).



An attempt to use a really null ideal Switch resistance (Roff=0) has been performed. At the time of the switch SW1_2 close at 3 s, an error occur (due to division by zero). This shows that it is mandatory in the model to have a not null circuit resistance.
One can mention, that in this particular case, the value of the energies in C1 and C2 after the switch closes (and at time of the zero division) are already at their expected values.



2. A more real model

A resistor R1_2 has been added between C1 and C2.





The results are the same as above, and the energy lost is now seen as Joule effect losses into the new resistor added : R1_2 with Ejoule2= 0.05J.

3. Conclusion

There are no paradox with EcosimPro for solving the "two capacitor paradox" because a full ideal circuit without resistance cannot be simulated. *A possible turnaround could be to add the equations for the electro-magnetic losses at the switch (from the Poynting vector)...*

The so called "ideal switch" is not so ideal because there are internal resistance. And without such resistance the model cannot run.

The major lesson is that charging a capacitor from a constant voltage source is a very bad design, whatever the resistance in the circuit, with an energy efficiency of only 50%.

Ref. Kirk T. McDonald, " A Capacitor Paradox, " Princeton, 2002,- 2013

Annex: Traceability

Listing of the experiment

```
/*-----------------------------------------------------------------------------------------
LIBRARY: MY_SAT
COMPONENT: ConvEnergyLossCst
PARTITION: default
EXPERIMENT: exp1
TEMPLATE: TRANSIENT
CREATION DATE: 11/12/2015
-----------------------------------------------------------------------------------------*/
-- ' 11/12/2015 17:02:12
EXPERIMENT exp1 ON ConvEnergyLossCstR1_2.default
DECLS
OBJECTS
INIT
  -- initial values for state variables
  C1.v = 0
  C2.v = 0
 BOUNDS
  -- Set equations for boundaries: boundVar = f(TIME;...)
  DC.s_in.signal[1] = 10
BODY
  -- creates an ASCII file with the results in table format
  -- REPORT_TABLE("results.rpt", "*")
  -- set the debug level (valid range [0,4])
  DEBUG_LEVEL= 1
  -- select default integration solver
  IMETHOD= DASSL
  -- Set flag to stop when bad numerical operation occurs (eg division by 0). By default do not stop.
  setStopWhenBadOperation(FALSE)
  -- set relative and absolute tolerance for DASSL solver (transient solver)
  REL_ERROR = 1e-006
  ABS_ERROR = 1e-006
  -- set relative tolerance for algebraics solver (steady solver)
  TOLERANCE = 1e-006
  -- when to report results
  REPORT_MODE = IS_STEP--CINT -- by default report at every CINT time or event detection
  R0.R=100
  R1_2.R=100
  SW0.b_fire.signal[1]=FALSE
  SW1_2.b_fire.signal[1]=FALSE
  SW0.b_fire.signal[1]=TRUE AFTER 1
  SW0.b_fire.signal[1]=FALSE AFTER 2
  SW1_2.b_fire.signal[1]=TRUE AFTER 3
  -- calculates a transient integration
  TIME = 0
  TSTOP = 5
  CINT = 0.001
  INTEG()
END EXPERIMENT
```

Listing of the model

```
-- Generated automatically by - EcosimPro - 5.4.16
USE ELECTRICAL VERSION "3.1.3"
-- EL code of the schematic.
-- The COMPONENT definition lines are blocked for edition.
-- You can edit the parameters clicking over them.
-- ' 11/12/2015 17:02:30
COMPONENT
ConvEnergyLossCstR1_2
DECLS
  REAL PowR0,PowR1_2
  REAL Ejoule1,Ejoule2, VR0,VR1_2,
VC1,VC2,dQ1,dQ2, Ecapa1,Ecapa2,Vin,
Ecapa1check, Ecapa2check
  REAL EDC, ER0C1,ER1_2C1C2, EjouleSW
TOPOLOGY

  ELECTRICAL.Capacitor C1(
   is_a C2)

  ELECTRICAL.Capacitor C2(
  C = 2000e-006 -- Non default value.
  )

  ELECTRICAL.Ground Ground_1

  ELECTRICAL.IdealSwitch SW0(
   is_a SW1_2)

  ELECTRICAL.IdealSwitch SW1_2(
  Roff = 1e-015, -- Non default value.
  Gon = 1e-015 -- Non default value.
  )

  ELECTRICAL.Resistor R0(
   is_a R1_2)

  ELECTRICAL.Resistor R1_2(
  R = 1)

  ELECTRICAL.VoltageSignal DC
  CONNECT C1.e_p TO R1_2.e_n
  CONNECT C2.e_n TO C1.e_n
  CONNECT Ground_1.e_p TO C1.e_n
  CONNECT SW0.e_n TO C1.e_p
  CONNECT SW1_2.e_n TO C2.e_p
  CONNECT R0.e_p TO SW0.e_p
  CONNECT R0.e_n TO DC.e_p
  CONNECT R1_2.e_p TO SW1_2.e_p
  CONNECT DC.e_n TO Ground_1.e_p
INIT
  Ejoule1=0
  Ejoule2=0
  EDC=0
  Ecapa1check=0
  Ecapa2check=0
CONTINUOUS
  Vin=DC.e_p.v-DC.e_n.v
  VR0 = R0.e_p.v-R0.e_n.v
  VR1_2 = R1_2.e_p.v-R1_2.e_n.v
  VC1 = C1.e_p.v-C1.e_n.v
  VC2 = C2.e_p.v-C2.e_n.v

EjouleSW'=SW1_2.e_p.i**2 * SW1_2.Roff
--Energy power supply
  EDC'=Vin* DC.e_p.i
--power in resistors
  PowR0=R0.e_p.i**2 * R0.R
  PowR1_2=R1_2.e_p.i**2 * R1_2.R
--integration of resistance  power = i V dt= R i² dt
  Ejoule1'=PowR0
  Ejoule2'=PowR1_2
--Energy formulae for capacitors (checked)
  Ecapa1=0.5*C1.C*VC1**2
  Ecapa2=0.5*C2.C*VC2**2
--integration of capacitor power = i V dt= dQ* V = C dV *  V
  dQ1=C1.C*VC1'
  dQ2=C2.C*VC2'
  Ecapa1check'= dQ1*VC1
  Ecapa2check'= dQ2*VC2
--Balances energies
  ER0C1=Ejoule1+Ecapa1
  ER1_2C1C2=Ejoule2+Ecapa1+Ecapa2
END COMPONENT
```